                                           4. Funktionen


4._____Funktionen
In  diesem  Abschnitt  werden die im  BASIC-System  BASI  1520 (SCPX)  vorhandenen  Standardfunktionen (ausser Funktionen  im Zusammenhang mit Datei - und Unterprogrammarbeit,  d.h. ausser EOF, LOC, LOF, INP, USR, PEEK und VARPTR) beschrieben.

Diese  Funktionen  koennen in einem BASIC-Programm  aufgerufen werden,  ohne dass sie definiert werden muessen. Die Argumente der Funktionen werden immer in Klammern gesetzt.

Bei  der Beschreibung der Funktionen werden die  Argumente  in der folgenden Form abgekuerzt:

X und Y   - beliebiger numerischer Ausdruck

I und J   - Integer - Ausdruecke

X$ und Y$ - Zeichenkettenausdruecke

Werden Funktionen im Programm verwendet, die einen Gleitkommawert liefern, wo eigentlich ein Integerwert gefordert wird, so rundet das BASIC-System den Wert.  Der sich ergebende Integerwert ist der Wert der Funktion.

Mit   dem  BASIC - Interpreter  werden  nur  Integerwerte  und Ergebnisse  mit  einfacher  Genauigkeit  von  den   Funktionen ausgegeben.  Funktionen  mit doppelter Genauigkeit werden  nur vom BASIC - Compiler unterstuetzt.

4.1.___Arithmetische_Funktionen

4.1.1._ABS
Format:      ABS(X)

Wirkung:     Ergibt den Absolutwert des Ausdruckes X

Beispiel:    PRINT ABS(3*(-9))

              27

             0k

4.1.2._INT
Format:      INT(X)

Wirkung:     Liefert die groesste Integerzahl <= X.

Beispiele:   PRINT INT(76.91)

              76

             0k

             PRINT INT(-26.33)

             -27

             0k

             Druck eines Kommas anstelle Punkt:

             30 N = 4562.32

             40 ? INT(N) CHR$(8) "," USING "##";

                INT(100*(N-INT(N)+.005))

             RUN

              4562,32

             0k

             Siehe  auch die FIX - und CINT - Funktionen,  die              auch Integerwerte ausgeben.

4.1.3._SGN
Format:      SGN(X)

Wirkung:     Wenn X>0, dann liefert SGN(X)  = 1.

             Wenn X=0, dann liefert SGN(X)  = 0.

             Wenn X<0, dann liefert SGN(X)  = -1.

Beispiel:    ON SGN(X) + 2  GOTO 100, 200, 300

             Verzweigung nach 100, wenn X negativ ist.

                              200, wenn X null ist.

                              300, wenn X positiv ist.

4.1.4._FIX
Format:      FIX(X)

Wirkung:     Liefert  den abgetrennten  Integer - Teil  von  X       

             FIX(X)  ist  aequivalent dem  SGN(X)*INT(ABS(X)).              Der  Hauptunterschied  zwischen FIX und  INT  ist              der, dass FIX nicht die naechst niedere Zahl fuer              das negative X liefert.

Beispiel:    PRINT FIX(37.75)

              37

             0k

             PRINT FIX(-37.75)

             -37

             0k

4.2.___Konvertierung_von_Zahlenwerten
4.2.1._CINT
Format:      CINT(X)

Wirkung:     Wandelt  X in eine Integerzahl durch Rundung  um.                Liegt  X nicht im Bereich -32768 bis 32767,  dann              wird ein Fehler "Overflow" angezeigt.

Beispiel:    PRINT CINT(22.78)

              23

             0k

Siehe die CDBL - und CSNG - Funktionen fuer das Umwandeln  von Zahlen  in Daten vom Typ einer doppelten Genauigkeit und einer einfachen  Genauigkeit.   

Siehe auch die FIX - und INT - Funktionen, die beide Integer - Zahlen ausgeben.

4.2.2._CSNG
Format:      CSNG(X)

Wirkung:     Wandelt X in eine Zahl mit einfacher  Genauigkeit                           um.

Beispiel:    10 C#=125.9871#

             20 PRINT C#;CSNG(C#)

             RUN

              125.9871  125.987

             0k

Siehe die CINT - und CDBL - Funktionen fuer das Umwandeln  von Zahlen in Daten  der Typen  Integer und  doppelte Genauigkeit.

4.2.3._CDBL
Format:      CDBL(X)

Wirkung:     Wandelt X in eine Zahl mit doppelter Genauigkeit              um.

Beispiel:    10 A=395.83

             20 PRINT A; CDBL(A)

             RUN

              395.83  395.8300170898438

             0k

4.3.___Exponential-Funktionen
4.3.1._EXP
Format:      EXP(X)

Wirkung:     Wandelt  X in e hoch X um.  X  muss  kleiner  als              87.3365  sein.   Ueberlauf  wird  angezeigt,  der              maximal darstellbare Wert wird mit dem  richtigen              Vorzeichen   als   Ergebnis  geliefert  und   die              Abarbeitung geht weiter.

Beispiel:    10 A=3.51

             20 PRINT EXP(A+1.1)

             RUN

              100.484

             0k

4.3.2._LOG
Format:      LOG(X)

Wirkung:     Liefert als Wert den natuerlichen Logarithmus von              X. X muss groesser Null sein.

Beispiel:    10 PRINT LOG(63*2)

             RUN

              4.83628

             0k

4.3.3._SQR
Format:      SQR(X)

Wirkung:     Liefert  als  Wert  die Quadratwurzel  von  X  in              einfacher  Genauigkeit.  X  muss groesser  gleich              Null sein.

Beispiel:    10 FOR I = 16 TO 32 STEP 4

             20 PRINT I, SQR(I)

             30 NEXT I

             RUN

              16            4

              20            4.47214

              24            4.89898

              28            5.2915

              32            5.65686

             Ok

4.4.___Trigonometrische_Funktionen

4.4.1._SIN
Format:      SIN(X)

Wirkung:     Berechnet  den  Sinus  von X  mit  einfacher  Ge             nauigkeit. X ist der Winkel in Bogenmass.

Beispiel:    10 PRINT SIN(1.321)

             RUN

              .968963

             0k

4.4.2._COS
Format:      COS(X)

Wirkung:     Berechnet   den  Cosinus  von  X  mit   einfacher                    Genauigkeit. X ist der Winkel in Bogenmass.

Beispiel:    10 X = 2*COS(.4)

             20 PRINT X

             RUN

              1.84212

             0k

             COS(X)=SIN(X+3.14159/2)

4.4.3._TAN
Format:      TAN(X)

Wirkung:     Berechnet  den  Tangens von X  mit einfacher  Ge-             nauigkeit.  X  ist der Winkel in Bogenmass.  

             Wird  der maximal darstellbare Wert  ueberschrit             ten,  erfolgt  Fehleranzeige.  Der  maximal  dar             stellbare  Wert mit dem richtigen Vorzeichen wird               als Funktionswert geliefert und die   Abarbeitung               fortgesetzt.

Beispiel:    10 Y=Q*TAN(X)/2

4.4.4._ATN
Format:      ATN(X)

Wirkung:     Berechnet den  Arkustangens von X und liefert als              Funktionswert den Winkel in Bogenmass im  Bereich              von -pi/2 bis +pi/2.

             Der   Ausdruck   X  kann  von  einem   beliebigen              numerischen  Typ sein,  aber die Abarbeitung  von              ATN   wird   immer   in   einfacher   Genauigkeit              durchgefuehrt.

Beispiel:    10 INPUT X

             20 PRINT ATN(X)

             RUN

             ? 3

              1.24905

             0k

4.5.___Zeichenketten-Funktionen
4.5.1._ASC
Format:      ASC(X$)

Wirkung:     Ergibt einen numerischen Wert, der dem ASCII-Code                    des  ersten  Zeichens  der  Kette  X$  entspricht              (siehe Anhang C fuer ASCII-Codes). 

             Wenn  X$  die leere Zeichenkette  ist,  wird  ein              Fehler "Illegal function call" ausgegeben.

Beispiel:    10 X$ = "TEXT" 

             20 PRINT ASC(X$)

             RUN

              84

             0k

             ? ASC("R")

              82

             0k

4.5.2._CHR$
Format:      CHR$(I)

Wirkung:     Der Wert von CHR$ ist das Zeichen,  das zum  Wert              des   Ausdrucks  gemaess  der  ASCII-Code-Tabelle              (Anlage C) gehoert. 

             I  ist  ein Wert zwischen 0/ und  255.  

             CHR$  wird  normalerweise  benutzt,  um  ein  be             stimmtes  Zeichen zum Bildschirm zu bringen. 

             Z.B.  koennte  CHR$(12) gesendet werden,  um  den              Bildschirm  zu  loeschen und den  Kursor  an  den              Anfang des Bildschirms zu bringen. Ebenso koennte              ein BEL - Zeichen (CHR$(7)) als Vorspann zu einer              Fehlermeldung gesendet werden.

Beispiel:    PRINT CHR$(66)

             B

             0k

4.5.3._HEX$
Format:      HEX$(X)

Wirkung:     Erzeugt eine Zeichenkette,  die den hexadezimalen              Wert des dezimalen Arguments darstellt.

             X  wird  zu  einer  Integerzahl  gerundet,  bevor              HEX$(X) erzeugt wird.

Beispiel:    10 INPUT "Dezimalzahl: ",Z

             20 Z$=HEX$(Z)

             30 PRINT  "Die Dezimalzahl ";Z;  "entpricht der                 Hexazahl ";Z$

             RUN

             Dezimalzahl: 33

             Die Dezimalzahl  33 entspricht der Hexazahl 21

             Ok

4.5.4._OCT$
Format:      OCT$(X)

Wirkung:     Erzeugt  eine Zeichenkette,  die den oktalen Wert              des  dezimalen  Arguments darstellt.  X  wird  zu              einer ganzen Zahl gerundet,  bevor OCT$(X) ausge             fuehrt wird.

Beispiel:    PRINT OCT$(25)

              31

             0k

4.5.5._STR$
Format:      STR$(X)

Wirkung:     Wandelt einen Zahlenwert in einen  Zeichenketten             wert

Beispiel:    10 INPUT "Eingabe Zahlen Z1,Z2:",Z1,Z2

             20 PRINT LEN(STR$(Z1)),LEN(STR$(Z2))

             RUN

             Eingabe Zahlen Z1,Z2:5132,81

              5             3

             0k

4.5.6._VAL
Format:      VAL(X$)

Wirkung:     Gibt  den  numerischen Wert der  Zeichenkette  X$              aus.

             Die VAL-Funktion trennt auch vorangestellte Leer             zeichen,  Tab's und Linefeeds (Zeilenschaltungen)              von der Argumentenkette X$ ab.

             z.B. gibt VAL(" -3") -3 aus anstatt von 0/.

Beispiel:    10 INPUT "Eingabe Zahlen X,Y:"X$,Y$

             20 LET Z=VAL(X$)+VAL(Y$)+50

             30 PRINT "X=";X$;" Y=";Y$;" X+Y=";Z

             RUN

             Eingabe Zahlen X,Y: 3,2

             X=3 Y=2 X+Y= 55

             0k

4.5.7._LEN
Format:      LEN(X$)

Wirkung:     Liefert  die  logische Laenge  der  Zeichenkette,              d.h.  die  Anzahl  der Zeichen von  X$,  die  zum              aktuellen  Wert gehoeren.  Auch Steuerzeichen und              Leerzeichen   koennen   dabei   zum   Wert    der              Zeichenkette  gehoeren,  d.h.  auch diese  werden              mitgezaehlt.

Beispiel:    10 X$= "Ein herrlicher Tag !"

             20 PRINT LEN(X$)

             RUN

              20 

             0k

4.5.8._INSTR
Format:      INSTR([I,]X$,Y$)

Wirkung:     Sucht  nach dem ersten Auftreten der Zeichenkette              Y$ in X$ und bringt die Position zurueck,  an der              diese gefunden wurde.

             Das  wahlweise I ist die Startposition,  von  der              aus die Suche beginnen soll.

             I  muss  sich im Bereich von 1 ...  255  bewegen. 

             Ist  I groesser als die Zahl der Zeichen  in  X$,              oder X$ hat den Wert leere Zeichenkette, oder der              Zeichenkettenwert  von Y$ in X$ wird nicht gefun             den, so liefert die Funktion den Wert Null.

             Wenn Y$ den Wert der leeren Zeichenkette hat,  so              liefert die Funktion den Wert 1.

Beispiel:    10 A$="DAS BASIC HANDBUCH"

             20 B$="S"

             30 PRINT INSTR(A$,B$);INSTR(5,A$,B$)

             RUN

              3  7

             0k

Beachte:     Wenn  I = 0 spezifiziert ist,  wird eine  Fehler             meldung "Illegal function call" ausgegeben.

4.5.9._LEFT$
Format:      LEFT$(X$,I)

Wirkung:     LEFT$  gibt die I ersten (linkesten) Zeichen  des              Zeichenkettenausdrucks zurueck.

             I gibt die Position des letzten  zurueckgegebenen              Zeichens an.

             I muss im Bereich 0 ... 255 liegen. 

             Ist I groesser als LEN(X$), dann wird der gesamte              Wert X$ geliefert.  Ist I=0/,  dann wird als  Wert              der Funktion die leere Zeichenkette geliefert.

Beispiel:    10 X$="BASIC-HANBUCH"

             20 Y$=LEFT$(X$,5)

             30 PRINT Y$

             RUN

             BASIC

             0k

4.5.10._RIGHT$
Format:      RIGHT$(X$,I)

Wirkung:     RIGHT$  gibt die rechten I-Zeichen (I-Zeichen von              rechts) von X$ als Wert zurueck. 

             Ist  I=LEN(X$),  wird  X$ komplett als  Wert  zu             rueckgegeben. 

             Ist  I=0/,   wird   die leere   Zeichenkette   als               Funktionswert geliefert.

Beispiel:    10 X$="DIE BASIC-ANWEISUNG"

             20 PRINT RIGHT$(X$,9)

             RUN

             ANWEISUNG

             0k

4.5.11._MID$
Format:      MID$(X$,I[,J])

Wirkung:     Erzeugt  eine Zeichenkette der Laenge J  Zeichen,              beginnend mit dem I. Zeichen.

             I und J muessen im  Bereich von 1 ... 255 liegen.

             Fehlt  J  oder  es  sind weniger  als  J  Zeichen                           rechts vom I-ten Zeichen vorhanden,  werden  alle              aeussersten   rechten  Zeichen,   beginnend   mit               dem  I-ten Zeichen, ausgegeben.

             Ist  I>LEN(X$),  ergibt  sich als Wert die  leere              Zeichenkette.

             Ist I=0/, wird die Fehlermeldung "Illegal function              call" ausgegeben.

Beispiel:    10 A$="EINE EINFACHE FUNKTION"

             20 PRINT "EIN ";MID$(A$,6,8);"S BEISPIEL"

             RUN

             EIN EINFACHES BEISPIEL

             0k

4.5.12._STRING$
Formate:     STRING$(I,J)

             STRING$(I,X$)

Wirkung:     Gibt  eine Zeichenkette der Laenge I  aus,  deren              Zeichen  alle ASCII-Code J oder das erste Zeichen              von X$ haben.

Beispiel:    10 A$=STRING$(10,61)

             20 PRINT A$;"BASIC-HANDBUCH";A$

             RUN

             ==========BASIC-HANDBUCH==========

             0k

4.5.13._SPACE$
Format:      SPACE$(X)

Wirkung:     Liefert als Wert X Leerzeichen.

             X wird auf eine Integerzahl gerundet und muss  im              Bereich von 0 bis 255 liegen.

Beispiel:    10 FOR K=1 TO 4

             20 A$=SPACE$(K)

             30 PRINT A$;K

             40 NEXT K

             RUN

               1

                2

                 3

                  4

             0k

Siehe auch die SPC-Funktion.

4.5.14._INKEY$
Format:      INKEY$

Wirkung:     Mit  dieser Funktion ist die Eingabe von  Zeichen              ueber  die  Tastatur moeglich,  ohne  <ENTER>  zu              bedienen. Wurde kein Zeichen eingegeben, wird  im              Programm fortgesetzt.

             Alle  Zeichen mit Ausnahme von  Control-C  werden              zum Programm uebergeben.

             (Control-C  beendet  das  Programm,  ausser  beim              Compiler, dort wird ^C auch weitergegeben.)

Beispiel:    10 A$=""

             20 FOR I=1 TO 1000    'Warteschleife fuer Eingabe

             30 B$=INKEY$

             40 A$=A$+B$

             50 NEXT I

             60 PRINT A$

             RUN

             ABC                   (ABC wurde in Zeile 30

             Ok                    eingegeben)

4.5.15._INPUT$
Format:      INPUT$(X)

Wirkung:     Mit dieser Funktion kann eine Zeichenkette von  X              Zeichen  eingegeben werden,  ohne <ENTER> zu  be             dienen.   Es  muessen  in jedem Falle  X  Zeichen              eingegeben werden. Das Programm wartet, bis diese              X Zeichen eingegeben sind.

             Alle  Steuerzeichen  mit Ausnahme  von  Control-C              werden  uebergeben.  ^C  wird benoetigt,  um  die              Ausfuehrung der INPUT$-Funktion zu unterbrechen.

Beispiel:    10 A$=INPUT$(2)          '2 Zeichen sind einzu-

             20 IF A$="EE" THEN 50    geben

             30 PRINT A$

             40 GOTO 10

             50 PRINT "ENDE"

             60 END

             RUN

             AB

             XY

             ENDE

             0k

4.6.___Dienstleistungsfunktionen
4.6.1._RND
Format:      RND[(X)]

Wirkung:     Liefert  eine Zufallszahl zwischen 0 und  1.  Mit              jeder Abarbeitung des Programms wird die  gleiche              Zufallszahlenfolge  ausgegeben,  ausser wenn  der              Zufallszahlengeneratorstart  ueber die RANDOMIZE-             Anweisung   veraendert  wird.   

             Auch wenn X<0/ spezifiziert wird,  wird immer  die              gleiche Folge fuer jedes angegebene X ermittelt.

             Wenn  X>0/  oder X  weggelassen  wurde,  wird  die              naechste Zufallszahl in der Folge erzeugt.

             X=0/ wiederholt die letzte erzeugte Zufallszahl.

Beispiel:    10 FOR K=1 TO 4

             20 PRINT INT(RND*100+2);

             30 NEXT K

             RUN

              26  32  33  53

             0k

4.6.2._FRE
Format:      FRE(X/)

             FRE(X$)

Wirkung:     FRE ermittelt die Anzahl der Bytes,  die vom Pro             gramm nicht belegt sind. 

             Die angegebenen Argumente sind Scheinargumente. 

             Es ist auch FRE("") moeglich.

             FRE("")  verdichtet den Speicher,  dies kann aber              bis zu 2 Minuten dauern ("Abfall-Sammlung").

             BASIC wird keine "Abfall-Sammlung" ausloesen, bis              der ganze verfuegbare Speicherraum benutzt wurde.              Deshalb   wird  eine  periodische  Benutzung  von              FRE("") dazu fuehren, dass jede "Abfall-Sammlung"              kuerzere Verzoegerungszeiten verursacht.

Beispiel:    PRINT FRE(1)

              27718

             0k

4.7.___Funktionen_zur_Drucker-/_Bildschirm-Steuerung
4.7.1._TAB
Format:      TAB(I)

Wirkung:     Verschiebt  den  Kursor / Druckkopf an  die  I-te              Position.

             Ueberschreitet die augenblickliche  Druckposition              bereits  I,  dann geht TAB zu dieser Position  in              der  naechsten Zeile.  TAB(1) ist die  aeusserste              linke  Position.  Die aeusserste rechte  Position              ist  die  Druckbreite minus 1   (siehe  Anweisung              WIDTH). 

             I muss im Bereich von 1 ... 255 liegen.  

             TAB  ist nur in PRINT- und LPRINT-Anweisungen  zu              verwenden.

Beispiel:    10 PRINT "ABC" TAB(20) "XYZ"

             20 PRINT

             30 PRINT "XXX" TAB(15) "111"

             RUN

             ABC                XYZ

             XXX           111

             0k

4.7.2._SPC
Format:      SPC(I)

Wirkung:     Verschiebt  den  Kursor / Druckkopf  um  I  Posi             tionen.

             SPC ist nur mit PRINT - oder LPRINT - Anweisungen              zu verwenden.

             I muss im Bereich von 0/ ... 255 liegen.

             Ein ";" kann der SPC(I)-Funktion folgen.

             Siehe auch die SPACE$-Funktion.

Beispiel:    10 LET A$="15": LET B$="999"

             20 PRINT A$ SPC(10) B$

             RUN

             15          999

             0k

4.7.3._LPOS
Format:      LPOS(X)

Wirkung:     Die  Funktion liefert die aktuelle  Position  des              Druckkopfes   innerhalb  des   Zeilendruckpuffers              (nicht die physische Position des Druckkopfes).

             X ist ein Scheinargument.

Beispiel:    10 LPRINT "ABCDEFG";

             15 PRINT LPOS(X)

             20 LPRINT LPOS(X)

             30 PRINT LPOS(X)

             RUN

              8

              1

             0k

4.7.4._POS
Format:      POS(X)

Wirkung:     Liefert   die   aktuelle   Kursor-Position.   Die              aeusserste linke Position ist die Position 1. 

             X ist ein Scheinargument.

Beispiel:    10 INPUT;A

             20 PRINT POS(X)

             30 PRINT "ABCD";

             40 PRINT POS(X)

             RUN

             ? 123456 9

             ABCD 5

             0k

                          - # -


