                                      3. Anweisungen


Beispiel 4:  10 I=5

             20 FOR I=1 TO I+5

             30 PRINT I;

             40 NEXT

             RUN

              1  2  3  4  5  6  7  8  9  10

             0k

             In   diesem  Beispiel  wird  der  Zyklus  zehnmal              durchgefuehrt. Der Endwert fuer die Zyklusvariab             le  wird immer eingestellt,  ehe der  Anfangswert              eingestellt wird.

3.6.7._IF...THEN[...ELSE]_und_IF...GOTO_[...ELSE]
Format 1:    IF<Ausdruck> [,] THEN{<Anweisung(en)>|<Zeilen-             nummer>} [ELSE{<Anweisung(en)>|<Zeilennummer>}]

Format 2:    IF<Ausdruck> GOTO <Zeilennummer>

             [ELSE{<Anweisung(en)>|<Zeilennummer>}]

Zweck:       In  Abhaengigkeit  vom Ergebnis eines  Ausdruckes              wird  das Abarbeiten  von Anweisungen ausgewaehlt.

Bemerkungen: Ist  das  Ergebnis  von  <Ausdruck>  wahr,   wird                           der THEN- oder GOTO-Teil ausgefuehrt. 

             THEN  koennen entweder eine Zeilennummer zum Ver             zweigen  oder  eine oder  mehrere  auszufuehrende              Anweisungen  folgen.  Der  Anweisung  GOTO  folgt              immer eine Zeilennummer.

             Ist das Ergebnis von <Ausdruck> falsch, dann wird              der THEN- oder GOTO-Teil ignoriert und der  ELSE-             Teil,  wenn  vorhanden,  bzw.  die nach IF...THEN              folgende Anweisung  abgearbeitet. Es wird mit der              naechsten ausfuehrbaren Anweisung fortgesetzt. 

             Ein Komma vor THEN ist moeglich.

             Verschachteln_der_IF-Anweisungen
             IF...THEN...ELSE-Anweisungen  koennen  verschach-             telt  werden.  Das Verschachteln  wird nur  durch              die Laenge der Zeile begrenzt.

             Zum Beispiel:

             IF A>B THEN PRINT "GROESSER" ELSE IF B>A

                    THEN PRINT "KLEINER" ELSE PRINT "GLEICH"

             ist  eine  legale Anweisung.  

             Wenn  die Anweisung nicht die gleiche Anzahl  von              ELSE- und THEN-Teilen enthaelt,  wird jedes  ELSE              an das naechste unangepasste THEN angepasst. 

             Zum Beispiel:

             IF A=B THEN IF B=C THEN PRINT "A=C"

                    ELSE PRINT "A<>C"

             A<>C wird nicht angezeigt, wenn A<>B.

Anmerkung:   Wenn  man IF verwendet,  um die  Gleichheit  fuer              einen  Wert  zu pruefen,  der das Ergebnis  einer              Gleitkommaberechnung ist,  ist zu beachten,  dass              die  interne Darstellung des Wertes  nicht  exakt              sein  koennte.  Deshalb sollte die Pruefung ueber              den Bereich erfolgen,  ueber den die  Genauigkeit              des Wertes variieren koennte. 

             Um  beispielsweise  eine  berechnete  Variable  A              gegen den Wert 1.0 zu pruefen, ist zu verwenden:

               IF ABS(A-1.0)<1.0E-6 THEN...

Beispiel 1:  100 IF(I<30) OR (I>10) THEN DB=1979-1:GOTO 500

             110 PRINT "FEHLER"

             In  diesem  Beispiel  ermittelt ein  Test,  ob  I                              groesser als 10 oder kleiner als 30 ist.  Wenn  I              in  diesem Bereich liegt,  wird DB berechnet  und              das  Programm bei Zeile 500 fortgesetzt.  Wenn  I              nicht  in diesem Bereich liegt,  wird die  Durch             fuehrung mit der Zeile 110 fortgesetzt.

Beispiel 2:  210 IF PORT THEN PRINT A$ ELSE LPRINT A$

             Diese Anweisung veranlasst, dass die Druckausgabe              entweder  zum  Bildschirm  oder  zum  Drucker  in              Abhaengigkeit vom Wert einer Variablen PORT geht.

             Wenn PORT Null ist, geht die Ausgabe zum Drucker,              sonst geht die Ausgabe zum Bildschirm.

Beispiel 3:  10 INPUT X,Y

             20 PRINT X,Y

             30 IF X>Y THEN 40 ELSE 60

             40 PRINT X,">",Y

             50 GOTO 70

             60 PRINT X,"<",Y

             70 END

             RUN

             ? 15,10

              15             10      

              15            >               10

             Ok

             RUN

             ? 15,20

              15             20

              15            <               20

             Ok

3.6.8._WHILE...WEND
Format:      WHILE <Ausdruck>

             .

             .

             [<Zyklusanweisungen>]

             .

             WEND

Zweck:       Ausfuehren  einer Serie von Anweisungen in  einem              Zyklus, solange eine gegebene Bedingung wahr ist.

Bemerkungen: Ist  <Ausdruck> nicht Null  (d.h.  wahr),  werden              <Zyklusanweisungen>    ausgefuehrt,    bis    man                           einer WEND-Anweisung begegnet.  Das  BASIC-System              kehrt  dann zur WHILE-Anweisung zurueck und  kon             trolliert  den <Ausdruck>.  Ist der Ausdruck noch              wahr,  wird der Prozess wiederholt. Wenn er nicht              wahr  ist,  wird mit der auf  die  WEND-Anweisung              folgenden Anweisung fortgesetzt.

             WHILE/WEND-Zyklen  koennen in jedem  Niveau  ver             schachtelt  werden.  Jedes WEND passt zum letzten              WHILE.  Eine  WHILE-Anweisung ohne  nachfolgendes              WEND  verursacht  einen  Fehler  "WHILE   without              WEND",  und  eine WEND-Anweisung ohne  vorheriges              WHILE   verursacht  einen  Fehler  "WEND  without              WHILE".

Beispiel:    100 ZAEHLER=3

             120 WHILE ZAEHLER

             121 FOR I=1 TO 5

             130 PRINT I,I^2,I^3

             131 NEXT I

             135 ZAEHLER=ZAEHLER-1

             150 WEND

             160 PRINT "ENDE"

             170 END

             RUN

              1            1             1

              2            4             8

              3            9             27

              4            16            64

              5            25            125

              1            1             1

              2            4             8

              3            9             27

              4            16            64

              5            25            125

              1            1             1

              2            4             8

              3            9             27

              4            16            64

              5            25            125

             ENDE

             0k

3.7.___Dimensionieren_von_Feldern

3.7.1._DIM
Format:      DIM <Liste der indizierten Variablen>

Zweck:       Legt die maximalen Werte fuer die Indizierung der              Feldvariablen  fest  und  ordnet  dementsprechend               den Speicher zu.

Bemerkungen: Wenn   ein  Feldvariablenname  ohne   eine   DIM-

             Anweisung verwendet wird,  kann der   Maximalwert              der Indizes 10 betragen. 

             Wenn ein Index verwendet wird,  der groesser  als              das  spezifizierte  Maximum  ist,  erscheint  der              Fehler "Subscript out of range". 

             Der  Mindestwert ist immer 0/,  wenn nicht  anders              mit der Anweisung OPTION BASE spezifiziert worden              ist (siehe Anweisung OPTION BASE).

             Die   DIM-Anweisung  stellt  alle  Elemente   des              spezifizierten  Feldes auf einen Anfangswert  von              Null ein.

             Die   DIM-Anweisung  kann  ueberall  im  Programm              stehen. Sie muss jedoch ausgefuehrt werden, bevor              auf das Feld zugegriffen wird.

             Die  maximale  Anzahl der Feldvariablen  wird  in              runde Klammern eingeschlossen.

             Die Laenge der Variablen wird nicht angegeben.

             Wird  versucht,  ein Feld neu zu  dimensionieren,                           ohne es vorher zu loeschen,  erscheint ein Fehler              "Duplicate Definition" (siehe Anweisung ERASE).

             Siehe auch Punkt 1.7.2. Feldvariable

Beispiele:   10 DIM A(30)    ' =31 Bereiche

             20 FOR I=0 TO 30

             30 READ A(I)

             40 NEXT I

             50 DATA . . . . . .

             100 DIM A$(2,10)

             110 FOR K=O TO 2

             115 INPUT B

             120 FOR J=0 TO 10

             130 READ A$(K,J)

             140 PRINT A$(K,J)

             150 NEXT J

             160 NEXT K

             170 DATA 0,1,2,3,4,5,6,7,8,9,10

             180 DATA 0,1,2,3,4,5,6,7,8,9,10

             190 DATA 0,1,2,3,4,5,6,7,8,9,10

             200 END

3.7.2._OPTION_BASE
Format:      OPTION BASE n

             wobei der Wert von n = 1 oder 0/ sein kann.

Zweck:       Der kleinste Wert fuer  die  Indizierung  der              Feldvariablen wird festgelegt.

Bemerkungen: Fehlt  die Angabe,  wird als Standard  =0/ an-             genommen.

             Wenn die Anweisung

                     OPTION BASE 1           

             ausgefuehrt wird, ist der niedrigste Wert = 1.

3.7.3._ERASE
Format:      ERASE <Liste der Feldvariablen>

Zweck:       Felder/Bereiche aus einem Programm loeschen.

Bemerkungen: Felder koennen neu dimensioniert werden, oder der              vorher  zugeordnete  Raum im Speicher  kann  fuer              andere Zwecke verwendet werden,  wenn die  Felder              mit ERASE geloescht werden.

Anmerkung:   Der  BASIC-Compiler_BASC_1520(SCPX)  unterstuetzt              ERASE nicht.

Beispiel:    450 ERASE A,B

             460 DIM B(39)

3.8.___Anwendereigene_Funktionsdefinition__DEF_FN
Format:      DEF FN<Name> [(<Parameterliste>)]=<Funktions-

             definition>

Zweck:       Definieren  und  Benennen einer vom Anwender  ge             schriebenen Funktion.

Bemerkungen: <Name>   muss  ein  legaler  Variablenname  sein.              Dieser Name,  dem ein FN vorausgeht, ist der Name              der Funktion.

             <Parameterliste>  ist aus jenen Variablennamen in              der  Funktionsdefinition   zusammengesetzt,   die              ersetzt werden sollen,  wenn die Funktion gerufen              wird.  Die  Positionen in der Liste werden  durch              Kommas getrennt.

             <Funktionsdefinition>  ist ein Ausdruck,  der die              Operation der Funktion durchfuehrt.  Sie ist  auf              eine  Zeile  begrenzt.   Variablennamen,  die  in              diesem Ausdruck erscheinen,  dienen nur dazu, die              Funktion  zu definieren.  Sie sind ausserhalb der              Funktion ungueltig.

             Ein  Variablenname,  der in einer  Funktionsdefi-             nition verwendet wird, kann in der Parameterliste              erscheinen  oder auch nicht.  Wenn er  erscheint,              wird  der Wert des Parameters  geliefert,  sobald              die Funktion gerufen wird.  Sonst wird der gegen             waertige Wert der Variablen verwendet.

             Die Anweisung kann numerische oder Zeichenketten             funktionen definieren.

             Wird  ein  Typ  im  Funktionsnamen  spezifiziert,                          wird der Wert des Ausdrucks zu jenem Typ  festge             legt. 

             Wird  im  Funktionsnamen ein Typ  spezififiziert,              aber der Argumenttyp passt nicht,  erscheint  ein              Fehler "Type mismatch".

             Eine DEF FN-Anweisung muss vor dem ersten  Aufruf              der  Anwenderfunktion definiert sein.  Wird  eine              Funktion  gerufen,  bevor  sie  definiert  wurde,              erscheint ein Fehler "undefined user function". 

             DEF  FN  ist in der  direkten  Betriebsart  nicht              ausfuehrbar.

Beispiel 1:  5 REM KURSORPOSITIONIERUNG  ZEILE  / SPALTE

             10 A$ ="PROGRAMM"

             20 B$ ="BEISPIEL"

             30 DEF FNC$(X,Y) =CHR$(27)+CHR$(X+127)+

                               CHR$(Y+127)

             40 PRINT FNC$(10,45)   A$

             50 PRINT FNC$(11,45)   B$

             A$ wird auf der Zeile 10,  Spalte  45,  angezeigt              und B$ auf Zeile 11, Spalte 45.

             In der Zeile 40 und 50 wird die Funktion gerufen.

Beispiel 2:  10 A$="PROGRAMM"

             20 B$="BEISPIEL"

             30 X = 10 : Y = X + 35

             40 DEF FNC$ =CHR$(27)+CHR$(X+127)+CHR$(Y+127)

             50 PRINT FNC$,A$,B$

             60 X = X + 1

             70 GOTO 50

             In  diesem  Fall  werden X und  Y  nicht  in  der              Parameterliste definiert.

3.9.___Programmueberlagerung

3.9.1._CHAIN
Format:      CHAIN [MERGE] <Dateiname>[,[<Zeilennummer exp>] 

                   [,ALL] [,DELETE <Bereich>]]

Zweck:       Nachladen  eines Programmes von der  Diskette  in              den  Speicher  und Starten sowie  Weitergabe  von              Variablen  aus dem laufenden Programm an das  ge             rufene Programm.

             Dabei   wird   das  gesamte   laufende   Programm              geloescht.

             Die Programmabarbeitung wird nicht unterbrochen.

Bemerkungen: <Dateiname>  ist  der Name  des  Programmes,  das                           gerufen  wurde.  Dieser ist in Anfuehrungsstriche              zu setzen.

             <Zeilennummer exp> ist eine Zeilennummer oder ein              Ausdruck,  der eine Zeilennummer in dem gerufenen              Programm bestimmt. Sie ist der Ausgangspunkt fuer              die Ausfuehrung des gerufenen Programms. Wenn die              Angabe weggelassen wurde,  beginnt die Durchfueh             rung bei der ersten Zeile.

             Die  COMMON-Anweiung  kann verwendet  werden,  um              Variable zu uebergeben.

             <Zeilennummer  exp> wird nicht von  einem  RENUM-             Kommando beeinflusst.

Beispiel 1:  10 REM DEMONSTRIEREN DER CHAIN-ANWEISUNG

             20 REM AUFZEICHNEN MODUL UNTER DEM NAMEN "TEST1" 

                MIT OPTION A

             30 DIM A$(2),B$(2)

             40 A$(1)= "VEB ROBOTRON"

             50 A$(2)= "BUCHUNGSMASCHINENWERK"

             80 CHAIN "TEST2",,ALL

             90 PRINT

             100 PRINT B$(1)

             110 PRINT

             120 PRINT B$(2) :PRINT

             130 END

             10 REM "DIM A$(2),B$(2)" DARF NUR EINMAL 

                VERWENDET WERDEN

             20 REM AUFZEICHNEN MODUL UNTER DEM NAMEN "TEST2" 

                MIT OPTION A

             30 COMMON A$(),B$()

             40 PRINT

             50 PRINT A$(1),A$(2)

             60 B$(1)="VORHERIGES PROGRAMM WURDE GELOESCHT"

             70 B$(2)="WERTE BLEIBEN ERHALTEN"

             80 CHAIN "TEST1",90

             90 END

             Mit  der ALL-Wahlmoeglichkeit wird jede  Variable              in   dem  laufenden  Programm  zu  dem  gerufenen              Programm    weitergegeben.    Wenn    die    ALL-             Wahlmoeglichkeit  weggelassen  wurde,   muss  das              laufende    Programm    eine     COMMON-Anweisung              enthalten,  um  die  Variablen  festzulegen,  die              weitergegeben werden.

             Die  MERGE-Wahlmoeglichkeit  erlaubt,   dass  ein              Unterprogramm  als  Ueberlagerung  in  ein BASIC-             Programm eingemischt werden kann.  Das heisst, es              wird eine MERGE-Operation mit dem  gegenwaertigen              Programm und dem gerufenen Programm durchgefuehrt              (ohne Programmunterbrechung). 

             Das gerufene Programm muss eine ASCII-Datei sein,              wenn es gemischt werden soll.

             Soll   das   ueberlagerte  Unterprogramm   wieder              geloescht  werden,  damit ein neues Unterprogramm              hineingemischt  werden kann,  so ist die  DELETE-             Wahlmoeglichkeit zu verwenden.

             Die Zeilennummern im <Bereich> werden vom  RENUM-             Kommando beeinflusst.

Beispiel 2:  10 REM Demonstrieren der CHAIN-Anweisung

             20 REM Aufzeichnen Modul unter dem Namen "TEST1" 

                mit Option A

             30 DIM A$(2),B$(2)

             40 A$(1)="VEB ROBOTRON"

             50 A$(2)="BUCHUNGSMASCHINENWERK"

             60 B$(1)=""

             70 B$(2)=""

             80 CHAIN "TEST2",,ALL

             90 PRINT

             100 PRINT B$(1)

             110 PRINT

             120 PRINT B$(2): PRINT

             130 END

             10 REM "DIM A$(2),B$(2)" darf nur einmal ver-

                wendet werden

             20 REM Aufzeichnen Modul unter dem Namen "TEST2" 

                mit Option A

             30 COMMON A$(),B$()

             40 PRINT

             50 PRINT A$(1); A$(2)

             60 B$(1)= "vorheriges Programm wurde geloescht"

             70 B$(2)= "Werte bleiben erhalten"

             80 CHAIN "TEST1",90

             90 END

Beispiel 3:  10 REM Dieses Programm demonstriert die CHAIN-

                Anweisung  mit der Option MERGE

             20 REM Aufzeichnen Modul unter den Namen 

                "HAUPTPRO"

             30 A$="HAUPTPROGRAMM"

             40 CHAIN MERGE "UP1",100,ALL

             50 END

             100 REM Aufzeichnen Modul unter dem Namen "UP1" 

                 mit Option A

             110 PRINT A$; "wurde ersetzt durch UP1."

             120 A$= "UP1"

             130 B$= "UP2"

             140 CHAIN MERGE "UP2",110, ALL, DELETE 100-150

             150 END

             100 REM Aufzeichnen Modul unter dem Namen "UP2" 

                 mit Option A

             110 PRINT A$; "wurde ersetzt durch "B$;"."

             120 END

Anmerkung:   Die  CHAIN-Anweisung mit der MERGE-Option  laesst                           die  Dateien offen und bewahrt die aktuelle  Ein             stellung der Anweisung OPTION BASE.

             Wird  die  MERGE-Option  nicht  genutzt,  bewahrt              CHAIN  nicht die Variablentypen oder vom Benutzer              definierte  Funktionen fuer die Verwendung  durch              das verkettete Programm.  Das heisst,  die Anwei             sungen DEFINT,  DEFSNG,  DEFDBL,  DEFSTR oder DEF              FN,  die  gemeinsam benutzte Variable  enthalten,              muessen in dem verketteten Programm neu  zugewie             sen werden.

             Der   BASIC-Compiler__BASC__1520(SCPX)   enthaelt               nicht  die Optionen

                  ALL, MERGE, DELETE und Zeilennummer exp.

             bei CHAIN.

             Hier lautet  die  Anweisung   CHAIN <Dateiname>.

             Wird die  Kompatibilitaet mit dem  BASIC-Compiler              notwendig,  wird empfohlen,  die Anweisung COMMON              zu verwenden, um Variable zu uebergeben.

             Ueberlagerungen sollten nicht verwendet werden.

             Wird  die  MERGE-Option  benoetigt,  sollten  be             nutzerdefinierte  Funktionen ausgefuehrt  werden,              bevor CHAIN MERGE-Anweisungen vorkommen.

             Ansonsten   sind  benutzerdefinierte   Funktionen              undefiniert, nachdem die MERGE-Anweisung komplett              ausgefuehrt wurde.

3.9.2._COMMON
Format:      COMMON <Liste der Variablen>

Zweck:       Uebergabe von Werten zwischen Programmsegmenten.

Bemerkungen: Die  COMMON-Anweisung wird in Verbindung mit  der              CHAIN-Anweisung   verwendet.   

             COMMON-Anweisungen   koennen  ueberall  in  einem              Programm erscheinen,  obwohl empfohlen wird,  sie              am Anfang des Programmes zu schreiben. 

             Dieselbe Variable kann nur in einer COMMON-Anwei             sung  auftreten.      

             Feldvariable  werden  durch Anfuegen von "()"  an              den Variablennamen spezifiziert. 

             Sollen alle Variablen uebergeben werden,  verwen             det  man CHAIN mit der ALL-Option und laesst  die              COMMON-Anweisung weg (siehe Anweisung CHAIN).

Anmerkung:   Der BASIC-COMPILER_BASC_1520(SCPX) verwendet eine              modifizierte Version der COMMON-Anweisung.

             Die  COMMON-Anweisung muss im Programm vor  jeder              ausfuehrbaren Anweisung erscheinen.

             Nicht ausfuerbare Anweisungen sind:

                   COMMON

                   DEFINT, DEFSNG, DEFDBL, DEFSTR

                   DIM

                   OPTION BASE

                   REM

                   %INCLUDE

             Feldvariable,   die   in  einer  COMMON-Anweisung              verwendet werden,  muessen in einer DIM-Anweisung              vereinbart werden.

             Die  Standardform  der  COMMON-Anweisung  bezieht 

             sich auf ein blank COMMON. Mit dem BASIC-Compiler              ist  es am besten,  alle blank COMMON-Anweisungen               in  einer einfachen INCLUDE-Datei einzulagern und              diese in jedem Programm  zu  verwenden  durch die              Anweisung %INCLUDE.

Beispiel:    MENU.BAS    10 %INCLUDE COMDEF  

                         .

                         1000 CHAIN "PROG1"

             PROG1.BAS   10 %INCLUDE COMDEF

                         .

                         2000 CHAIN "MENU"

             COMDEF.BAS  100 DIM A(100),B$(200)

                         110 COMMON I,J,K,A()

                         120 COMMON A$,B$(),X,Y,Z

3.10.___Fehlerbehandlung

3.10.1._ERROR
Format:      ERROR <Integerausdruck>

Zweck:       - Um   das   Auftreten  eines  BASIC-Fehlers   zu                               simulieren   

             - Um Fehlercodes durch den Benutzer zu definieren

Bemerkungen: Der  Wert des <Integerausdruckes> muss zwischen 0              und 255 liegen. 

             Wenn  der  Wert  des  <Integerausdruckes>  gleich              einem Fehlercode ist,  der vom BASIC-System schon              benutzt  wird  (siehe Anhang  A),  simuliert  die              ERROR-Anweisung das Auftreten jenes Fehlers,  und              die entsprechende Fehlernachricht wird angezeigt.

Beispiel:    10  A = 5 : B = 3

             20  C = A+ B

             30  ERROR C

             40  END

             RUN

             undefined line number in 30

             oder in der direkten Betriebsart:

             ERROR 8          (Diese Zeile wird eingegeben)

             undefined  line  (Das BASIC-System   zeigt  diese                               Zeile an)

             Soll ein  eigener  Fehlercode  definiert  werden,             ist  ein Wert zu verwenden,  der groesser als der              fuer die BASIC-Fehlercodes vorgesehene Wert ist. 

             Es werden bevorzugt die hoechsten zur  Verfuegung              stehenden Werte verwendet, so dass die Kompatibi             litaet  beibehalten  wird,  wenn  zu  BASIC  noch 

             Fehlercodes  hinzugefuegt werden.  Dieser vom Be             nutzer definierte Fehlercode kann dann bequem  in              einer Fehlerbehandlungsroutine verwendet werden.

             Wenn    eine    ERROR-Anweisung    einen     Code              spezifiziert,   fuer  den  keine  Fehlernachricht              definiert wurde,  antwortet das BASIC-System  mit              der Fehlernachricht "Unprintable error".

             Eine  ERROR-Anweisung,  fuer die keine  Fehlerbe             handlungsroutine  vorhanden ist,  verursacht  die              Anzeige einer Fehlernachricht. Das Abarbeiten des              Programmes wird angehalten.

Beispiel:    10  ON ERROR GOTO 150

             20  INPUT A

             30  IF A<0 THEN ERROR 200

             40  IF A=0 THEN ERROR 210

             150 IF ERR=200 THEN PRINT A

             160 IF ERL=30  THEN RESUME 20

3.10.2._ERR-UND_ERL-VARIABLE
Format:      ERR = <Fehlercode>

             ERL = <Zeilennummer>

Zweck:       Die ERR- und ERL-Variablen werden gewoehnlich  in              IF...THEN-Anweisungen  verwendet,   um  den  Pro             grammfluss  in  die  Fehlerbehandlungsroutine  zu              lenken.

Bemerkungen: Wird  ein Unterprogramm zur Fehlerbehandlung ein             gefuegt,  enthaelt  die Variable ERR  den Fehler             code  fuer  den Fehler und die Variable  ERL  die              Zeilennummer  der Zeile,  in der der Fehler  ent             standen ist. 

             War die Anweisung, die den Fehler verursacht hat,              eine Anweisung der direkten Betriebsart, enthaelt              ERL 65535. 

             Um  zu pruefen,  ob ein  Fehler in einer direkten              Anweisung aufgetreten ist, ist zu programmieren:

                      IF 65535 = ERL THEN ...

             Sonst ist zu programmieren:

                      IF ERR = Fehlercode THEN ...

                      IF ERL = Zeilennummer THEN ...

             Befindet  sich  die Zeilennummer  nicht  auf  der              rechten  Seite  der Anweisung,  kann   sie  durch              RENUM nicht neu numeriert werden.

             Weil  ERL  und  ERR  reservierte  Variable  sind,              duerfen diese in einer LET-Anweisung nicht  links              vom Gleichheitszeichen erscheinen.

             Die Fehlercodes des BASIC-Systems BASI 1520(SCPX)              sind im Anhang A aufgelistet.

Beispiel:    siehe Anweisung ERROR

3.10.3._ON_ERROR_GOTO
Format:      ON ERROR GOTO <Zeilennummer>

Zweck:       Um  bei  Auftreten eines  Fehlers  die  1.  Zeile              der Fehlerbehandlungsroutine zu spezifizieren.

Bemerkungen: Erscheint  ein Fehler,  bewirken alle Fehler ein             schliesslich  der  Fehler,  die in  der  direkten              Betriebsart  entstanden sind  (wie  z.B.  Syntax             fehler),  einen  Sprung  zu einer  spezifizierten              Fehlerbehandlungsroutine.

             Existiert <Zeilennummer> nicht,  wird ein  Fehler              "Undefined line" erzeugt.

             Um   die  Fehlersuche   abzuschalten,  wird   ein 

             ON_ERROR_GOTO_0/ programmiert.  Bei  nachfolgenden              Fehlern  wird  die Systemfehlermeldung  oder  die              vereinbarte  Fehlermeldung gedruckt.  Das  BASIC-             System kehrt zum Befehlsniveau zurueck.

             Die  Fehlerbehandlungsroutine muss unbedingt  mit              RESUME <Zeilennummer> abgeschlossen werden.

Anmerkung:   Tritt ein Fehler waehrend der Ausfuehrung  einer              Fehlerbehandlungsroutine auf,  wird eine  Fehler             nachricht  gedruckt und die Ausfuehrung  beendet.              Es  ist keine Fehlersuche innerhalb einer Fehler             behandlungsroutine moeglich.

Beispiel:    30  A = 10

             40  B = 20

             50  ON ERROR  GOTO 130

             60  INPUT A1

             70  IF A1>A THEN ERROR 100

             80  PRINT A

             85  ON ERROR GOTO 150

             90  INPUT B1

             100 IF B1>B THEN ERROR 110

             110 PRINT B

             120 END

             130 A = A+1

             140 RESUME 60

             150 B = B+1

             160 RESUME 90

3.10.4._RESUME
Formate:     RESUME

             RESUME 0/

             RESUME NEXT

             RESUME <Zeilennummer>

Zweck:       Fortsetzung  der  Programmabarbeitung nach  einer              Fehlerbehandlungsroutine.

Bemerkungen: Es  kann  jedes der vier oben  gezeigten  Formate              verwendet  werden in Abhaenigkeit davon,  wo  die              Ausfuehrung wiederaufgenommen werden soll:

             RESUME               Die Ausfuehrung wird bei der              oder                 Anweisung  wieder  aufgenom             RESUME 0/             men,  die  den Fehler verur- 

                                  sacht hat.

             RESUME NEXT          Die Ausfuehrung wird bei der                                   Anweisung wiederaufgenommen,                                   die   der  fehlerhaften  An-                                  weisung unmittelbar folgt.

             RESUME<Zeilennummer> Die  Ausfuehrung  wird   bei                                                   der  <Zeilennummer>  wieder-                                                   aufgenommen.

             Eine RESUME-Anweisung, die nicht in einer Fehler             behandlungsroutine  auftritt,   verursacht   eine 

             Fehlernachricht "RESUME without error".

Beispiel:    Siehe Anweisung "ON ERROR GOTO"

                          - # -


